

GUIDE COMPLET DE CONFIGURATION DU HOTSPOT MIKROTIK

De A à Z

Rédigé par :

VOUFFO Wilfrid Beongel

MBOGOL Yves Bertrand

ASSONTIA BOUALI Yan

Version validée sur RouterOS 7.19.5 — MikroTik hAP ax² (C52iG-5HaxD2HaxD)

Sommaire

- Avertissement de compatibilité
- 1. Introduction et périmètre
- 2. Architecture réseau cible
- 3. Prérequis matériel et logiciel
- 4. Partie A — Configuration du routeur MikroTik
- 5. Partie B — Configuration du point d'accès Wi-Fi
- 6. Procédures d'importation et de restauration
- 7. Évolutions et perspectives du projet
- 8. Conclusion

Avertissement de compatibilité

Attention — Ce guide a été conçu, testé et validé exclusivement sur RouterOS version 7.19.5, sur un routeur MikroTik modèle hAP ax² (référence C52iG-5HaxD2HaxD). La compatibilité stricte n'est pas garantie sur des versions antérieures ou ultérieures de RouterOS, ni sur d'autres modèles de routeurs.


Nous recommandons, avant toute mise en production sur un matériel ou une version différente, de vérifier la disponibilité des commandes et propriétés utilisées dans ce document (notamment celles liées au mode d'appareil, au Hotspot Manager et aux interfaces Wi-Fi intégrées), certaines évoluant d'une version de RouterOS à l'autre. Nous conseillons également de tester l'intégralité de la procédure sur un routeur de test avant tout déploiement définitif.

1. Introduction et périmètre

Ce document constitue le guide de référence pour la configuration complète d'un hotspot Wi-Fi avec portail captif, bâti autour d'un routeur MikroTik RouterOS. Nous y détaillons, de façon méthodique et reproductible, l'ensemble des étapes ayant permis d'aboutir à la configuration finale validée de notre projet : accès utilisateur via portail captif, administration du réseau, et automatisation des tâches de maintenance.

Ce guide s'adresse à tout ingénieur ou technicien réseau disposant d'un routeur MikroTik compatible et d'un point d'accès Wi-Fi, et souhaitant reproduire l'intégralité de notre installation pour obtenir un système immédiatement fonctionnel.

2. Architecture réseau cible



Profils Hotspot : etudiant (2M/2M) • invite (512k/512k)

Élément	Interface / rôle	Adresse
Liaison montante (WAN)	ether1	Adresse fournie par le fournisseur d'accès
Réseau local du hotspot	bridge (regroupe ether2 à ether5, wifi1, wifi2)	192.168.1.1/24
Plage DHCP distribuée aux clients	bridge	192.168.1.10 à 192.168.1.254
Point d'accès Wi-Fi externe	branché sur un port du bridge, en mode pont	adresse attribuée par le DHCP du routeur

3. Prérequis matériel et logiciel

- Un routeur MikroTik RouterOS 7.19.5 (ou version compatible vérifiée)
- Un point d'accès Wi-Fi externe compatible mode pont (testé avec un modèle Ubiquiti NanoStation)
- WinBox installé sur le poste d'administration
- Un accès physique initial au routeur pour la première configuration
- Câblage Ethernet approprié entre le routeur, le point d'accès, et le poste d'administration

4. Partie A — Configuration du routeur MikroTik

Chaque étape ci-dessous précise son objectif technique, la procédure via l'interface graphique WinBox, puis son équivalent strict en ligne de commande.

4.1 Activation du mode avancé de l'appareil

Objectif : sur le matériel neuf, RouterOS restreint par défaut certaines fonctions professionnelles, dont le Hotspot Manager, en mode « home ». Il est nécessaire de passer en mode « advanced » avant toute autre configuration.

Procédure WinBox

Non disponible directement dans les menus graphiques : cette opération s'effectue exclusivement via le terminal (New Terminal).

Commande CLI

```
/system/device-mode/update mode=advanced
```

Attention — Cette commande requiert une confirmation physique : appuyez brièvement sur le bouton reset du routeur dans les 5 minutes suivant la commande. Le routeur redémarre ensuite avec le mode activé.

4.2 Constitution du bridge réseau

Objectif : regrouper l'ensemble des interfaces destinées aux clients (ports Ethernet et Wi-Fi intégré) au sein d'une interface unique, afin que le hotspot et le DHCP s'appliquent uniformément à tous les points d'entrée du réseau local.

Procédure WinBox

Menu Bridge → onglet Ports → vérifier que ether2, ether3, ether4, ether5, wifi1 et wifi2 apparaissent bien comme membres du bridge nommé « bridge ». L'interface ether1 doit rester en dehors, dédiée à la liaison montante.

Commande CLI (vérification)

```
/interface bridge port print
```

4.3 Adressage IP du réseau local

Procédure WinBox

Menu IP → Addresses → Nouveau : Address = 192.168.1.1/24, Interface = bridge.

Commande CLI

```
/ip address add address=192.168.1.1/24 interface=bridge
```

4.4 Serveur DHCP

Objectif : attribuer automatiquement une adresse IP, une passerelle et un serveur DNS à chaque appareil se connectant au réseau local.

Procédure WinBox

Menu IP → Pool → créer une plage (192.168.1.10 à 192.168.1.254), puis IP → DHCP Server → Nouveau, en associant l'interface bridge et le pool créé, puis onglet Networks pour définir la passerelle et le DNS.

Commande CLI

```
/ip pool add name=dhcp_pool3 ranges=192.168.1.10-192.168.1.254
/ip dhcp-server add address-pool=dhcp_pool3 interface=bridge name=dhcp1
/ip dhcp-server network add address=192.168.1.0/24 dns-server=192.168.1.1
gateway=192.168.1.1
```

4.5 Configuration DNS

Procédure WinBox

Menu IP → DNS : cocher « Allow Remote Requests », renseigner les serveurs 8.8.8.8 et 1.1.1.1 comme résolveurs.

Commande CLI

```
/ip dns set allow-remote-requests=yes servers=8.8.8.8,1.1.1.1
```

4.6 Translation d'adresses (NAT)

Objectif : permettre aux clients du réseau local, dont les adresses sont privées, d'accéder à Internet via l'adresse publique du routeur.

Procédure WinBox

Menu IP → Firewall → onglet NAT → Nouveau : Chain = srcnat, Out. Interface = ether1, onglet Action = masquerade.

Commande CLI

```
/ip firewall nat add action=masquerade chain=srcnat out-interface=ether1 comment="NAT WAN"
```

Remarque — Une seule règle de masquerade suffit et doit être conservée ; toute règle redondante ciblant le même trafic doit être supprimée afin de conserver une configuration lisible et facilement maintenable.

4.7 Règles de pare-feu

Objectif : autoriser le trafic légitime tout en filtrant les connexions non désirées, sur les chaînes input (trafic à destination du routeur) et forward (trafic traversant le routeur).

Commande CLI — chaîne input

```
/ip firewall filter add action=accept chain=input connection-
state=established,related
/ip firewall filter add action=drop chain=input connection-state=invalid
/ip firewall filter add action=accept chain=input in-interface=bridge
/ip firewall filter add action=drop chain=input
```

Commande CLI — chaîne forward

```
/ip firewall filter add action=accept chain=forward comment="FWD OK : reponses"
connection-state=established,related
/ip firewall filter add action=accept chain=forward comment="FWD OK : LAN vers WAN"
in-interface=bridge out-interface=ether1
/ip firewall filter add action=drop chain=forward comment="FWD BLOQUE : invalid"
connection-state=invalid
/ip firewall filter add action=drop chain=forward comment="FWD BLOQUE : reste"
```

Attention — L'ordre des règles est impératif. Une règle de blocage général placée avant une règle

d'autorisation empêche cette dernière de jamais s'appliquer.

4.8 Activation du Hotspot Manager

Procédure WinBox

Menu IP → Hotspot → Server Profiles → Nouveau : DNS Name = hotspot.local, Hotspot IP = 192.168.1.1. Puis onglet Servers → Nouveau : Interface = bridge, Address Pool = dhcp_pool3, Profile = hsprof1.

Commande CLI

```
/ip hotspot profile add dns-name=hotspot.local hotspot-address=192.168.1.1
name=hsprof1
/ip hotspot add address-pool=dhcp_pool3 disabled=no interface=bridge name=hotspot1
profile=hsprof1
```

4.9 Profils utilisateurs

Objectif : définir deux catégories d'utilisateurs avec des politiques de bande passante, de durée de session et de nombre d'appareils simultanés différenciées.

Paramètre	Profil « étudiant »	Profil « invite »
Débit (rx/tx)	2 Mbps / 2 Mbps	512 kbps / 512 kbps
Appareils simultanés	2	1
Durée de session	5 minutes	5 minutes
Délai de réauthentification	30 secondes	30 secondes

Commande CLI

```
/ip hotspot user profile add name=etudiant rate-limit=2M/2M shared-users=2 session-
timeout=5m mac-cookie-timeout=30s
/ip hotspot user profile add name=invite rate-limit=512k/512k shared-users=1 session-
timeout=5m mac-cookie-timeout=30s
```

4.10 Personnalisation des pages du portail captif

Trois fichiers HTML remplacent les pages par défaut du Hotspot Manager, déposés dans le dossier Files → hotspot du routeur : login.html (page de connexion), status.html (page affichée après authentification, indiquant IP, adresse MAC, volume de données échangées, durée de connexion), et logout.html (page de confirmation de déconnexion).

Procédure WinBox

Menu Files → double-clic sur le dossier hotspot pour l'ouvrir, puis glisser-déposer chaque fichier corrigé depuis le poste d'administration en écrasant les fichiers existants du même nom.

4.11 Création des comptes utilisateurs

Création manuelle

```
/ip hotspot user add name=<identifiant> password=<mot_de_passe>
profile=<etudiant_ou_invite>
```


Génération automatique d'un identifiant et d'un mot de passe

Afin d'éviter toute saisie manuelle répétitive et tout risque de doublon, le script suivant génère un identifiant et un mot de passe aléatoires de quatre caractères chacun, et gère le cas d'une éventuelle collision de nom :

```
:local login [:rndstr 4]
:local pass [:rndstr 4]
:local profilChoisi "etudiant"
:do {
  /ip hotspot user add name=$login password=$pass profile=$profilChoisi
} on-error={
  :put "Collision detectee, relancer le script"
}
:put ("Identifiant : $login / Mot de passe : $pass")
```

4.12 Expiration calendaire des comptes à 30 jours

Objectif : garantir qu'un compte reste utilisable sans restriction pendant trente jours à compter de sa création, quel que soit le volume d'utilisation, puis soit automatiquement rendu inutilisable, indépendamment du fait qu'il ait été ou non utilisé durant cette période.

Chaque nouveau compte doit être accompagné de la création d'une tâche planifiée unique, programmée à la date de création augmentée de trente jours. Nous avons retenu la suppression du compte (plutôt que sa simple désactivation) afin de libérer immédiatement l'identifiant pour une éventuelle réattribution future.

Commande CLI (exemple pour un compte nommé « vouffo », créé le 2 août 2026)

```
/system scheduler add name=expire-vouffo start-date=sep/01/2026 start-time=00:00:00
interval=0 on-event="/ip hotspot user remove [find name=\"vouffo\"]"
```

Conseil — Le nom de la tâche planifiée (expire-<identifiant>) doit impérativement reprendre l'identifiant exact du compte concerné, afin de conserver une correspondance claire et d'éviter toute tâche orpheline ou mal ciblée.

4.13 Sécurisation des services d'administration

Objectif : n'autoriser l'accès administrateur qu'aux seuls services et interfaces strictement nécessaires.

Commande CLI

```
/ip service disable www
/tool mac-server set allowed-interface-list=LAN
/tool mac-server mac-winbox set allowed-interface-list=LAN
```

Remarque — Le service www (interface d'administration web WebFig) doit rester désactivé, car il entre en conflit direct avec le portail captif : les deux services tentent d'utiliser le port 80 de la même adresse IP.

4.14 Sauvegardes automatiques quotidiennes

Objectif : garantir la disponibilité d'une sauvegarde récente de la configuration, sans intervention manuelle quotidienne.

Commande CLI

```
/system scheduler add interval=1d name=backup-quotidien on-event={
  :local d [/system clock get date]
  /system backup save name=("backup-" . $d)
```

```
/export file=("config-" . $d)
}
```

La récupération de ces fichiers sur le poste de l'administrateur s'effectue manuellement, via Files dans WinBox, par glisser-déposer, à la fréquence jugée nécessaire.

5. Partie B — Configuration du point d'accès Wi-Fi

Cette section présente les principes généraux de configuration d'un point d'accès Wi-Fi externe destiné à relayer la connexion vers le portail captif du routeur MikroTik. Notre déploiement a été réalisé avec un point d'accès Ubiquiti NanoStation ; les principes exposés ci-dessous restent applicables, dans leur logique générale, à tout point d'accès du marché disposant d'un mode pont (Grandstream, Ubiquiti, TP-Link, etc.).

5.1 Raccordement physique

Le point d'accès est relié, via son injecteur d'alimentation (PoE) le cas échéant, à un port membre du bridge du routeur MikroTik (ether2, ether3, ether4 ou ether5).

5.2 Adressage du point d'accès

Le point d'accès doit être configuré en client DHCP, afin de recevoir automatiquement une adresse dans la plage 192.168.1.10 à 192.168.1.254 attribuée par le routeur MikroTik. L'adresse attribuée peut être consultée depuis WinBox, dans Ip → DHCP Server → Leases.

5.3 Mode de fonctionnement — point impératif

Attention — Le point d'accès doit impérativement être configuré en mode Pont (Bridge / Access Point), et jamais en mode Routeur. Un point d'accès en mode routeur effectuerait sa propre distribution d'adresses IP et son propre NAT, ce qui empêcherait les appareils connectés en Wi-Fi d'être correctement pris en charge par le portail captif du routeur MikroTik.

5.4 Désactivation du DHCP local du point d'accès

Le service DHCP intégré au point d'accès, lorsqu'il existe, doit être désactivé sans exception, le routeur MikroTik demeurant l'unique distributeur d'adresses IP du réseau.

5.5 Diffusion du réseau Wi-Fi

Un identifiant réseau (SSID) et un mot de passe Wi-Fi (sécurité WPA2 ou WPA3) sont définis directement dans l'interface de configuration du point d'accès. Aucune authentification supplémentaire n'est nécessaire à ce niveau : l'authentification des usagers est intégralement prise en charge, en aval, par le portail captif du routeur MikroTik.

5.6 Gestion des VLAN

Notre déploiement n'a pas nécessité de segmentation par VLAN, l'ensemble du trafic transitant sur un unique réseau local (192.168.1.0/24). Pour un déploiement nécessitant plusieurs réseaux Wi-Fi isolés (par exemple un réseau administrateur distinct du réseau usagers), la création de VLAN dédiés, tant côté point d'accès que côté bridge MikroTik, constituerait une évolution pertinente, non mise en œuvre dans la présente configuration.

6. Procédures d'importation et de restauration

Deux méthodes distinctes permettent de restaurer une configuration sur un routeur MikroTik, chacune avec un usage et des contraintes propres.

6.1 Importation d'un fichier .rsc sur un routeur vierge

Le fichier .rsc est un export texte de commandes, exécutables séquentiellement. Il est portable entre différentes architectures matérielles et versions de RouterOS, sous réserve de compatibilité des commandes utilisées.

Procédure

- Réinitialiser le routeur à sa configuration d'usine si nécessaire
- Copier le fichier .rsc dans Files, via glisser-déposer depuis WinBox
- Exécuter, depuis New Terminal :

```
/import file=<nom_du_fichier>.rsc
```

- Vérifier, à l'issue de l'exécution, l'absence d'erreur de syntaxe et la bonne application de chaque section (adressage, hotspot, profils, comptes)

Remarque — Une propriété non reconnue par la version de RouterOS cible peut interrompre l'importation d'une ligne précise sans nécessairement bloquer l'ensemble du script. Il convient de toujours vérifier, après importation, que la configuration obtenue correspond intégralement à celle attendue.

6.2 Restauration d'un fichier .backup

Le fichier .backup est une image binaire complète du système, incluant la configuration et certains éléments internes. Sa restauration n'est fiable qu'entre routeurs de même architecture matérielle et de version RouterOS identique ou immédiatement compatible.

Procédure

- Vérifier au préalable l'architecture et la version du routeur cible :

```
/system resource print
```

- Copier le fichier .backup dans Files du routeur cible
- Effectuer un clic droit sur le fichier, puis sélectionner Restore
- Confirmer : le routeur redémarre avec la configuration restaurée, remplaçant intégralement la configuration existante

6.3 Différence d'usage entre les deux méthodes

Critère	.rsc (export)	.backup (sauvegarde complète)
Portabilité	Élevée, multi-architecture	Limitée à l'identique ou au très proche
Contenu	Commandes de configuration uniquement	Image système complète
Usage recommandé	Déploiement sur un nouveau matériel	Restauration après incident sur le même modèle
Fichiers du portail captif	Non inclus (à transférer séparément)	Non inclus (à transférer séparément)

7. Évolutions et perspectives du projet

Plusieurs fonctionnalités ont été étudiées au cours de ce projet sans être intégrées à la configuration finale, par choix de priorisation ou par manque de temps. Nous les présentons ici avec leur rôle fonctionnel et la logique de mise en œuvre envisagée, à destination de toute équipe souhaitant poursuivre ce travail.

7.1 Système de gestion et d'envoi d'alertes

Rôle fonctionnel : informer automatiquement l'administrateur, par courrier électronique ou message texte, en cas de panne ou d'anomalie détectée sur le réseau (perte de la liaison montante, indisponibilité du point d'accès, saturation de la bande passante, tentatives d'intrusion, espace de stockage insuffisant, échec d'une sauvegarde programmée), sans que celui-ci ait à surveiller le système en continu.

Logique de mise en œuvre : un outil de surveillance interne à RouterOS (Netwatch) vérifie à intervalle régulier l'état d'un service ou d'une interface, et déclenche un script en cas de changement d'état. Ce script journalise l'évènement et envoie une notification via un serveur de messagerie configuré au préalable. Exemple, pour la surveillance de la liaison montante :

```
/tool e-mail set server=smtp.gmail.com port=587 from="hotspot@domaine.org"
user="compte@gmail.com" password="mot_de_passe_application" tls=yes

/tool netwatch add host=8.8.8.8 interval=1m down-script={
:local msg "Perte de connexion Internet detectee."
/log error $msg
/tool e-mail send to="admin@domaine.org" subject="[HOTSPOT] Coupure WAN" body=$msg
}
```

7.2 Génération automatique de logins et de mots de passe

Rôle fonctionnel : permettre la création rapide et sécurisée de nouveaux comptes usagers, sans intervention de saisie manuelle et sans risque de doublon, au moment où un usager sollicite et règle son forfait d'accès.

Logique de mise en œuvre : un script RouterOS génère, à l'aide de la fonction interne de génération de chaînes aléatoires, un identifiant et un mot de passe, puis crée directement le compte hotspot avec le profil approprié. Une gestion d'erreur permet de relancer automatiquement la génération en cas de collision avec un identifiant déjà existant. Le résultat peut être communiqué à l'utilisateur sous forme de ticket imprimé.

7.3 Export automatisé des statistiques d'usage

Rôle fonctionnel : permettre à l'administrateur d'analyser, dans un tableur, l'évolution de la fréquentation du hotspot (répartition par profil, heures de forte affluence, tendance sur la durée), à des fins de dimensionnement et de justification du déploiement.

Logique de mise en œuvre : un script mesure, à intervalle régulier, le nombre de sessions actives par profil, et ajoute une ligne à un fichier au format CSV, directement exploitable dans un tableur. Ce script a été développé et validé manuellement au cours du projet ; son automatisation complète nécessite l'ajout d'une tâche planifiée déclenchant son exécution toutes les quinze minutes :

```
/system scheduler add name=sched-collecte interval=15m on-event=collecte-stats
```

7.4 Accès administrateur à distance

Rôle fonctionnel : permettre à l'administrateur de gérer le réseau sans présence physique sur site, condition particulièrement pertinente pour un déploiement en zone isolée.

Logique de mise en œuvre : un tunnel chiffré de type WireGuard, établi entre le routeur et un ou plusieurs postes administrateurs autorisés, permet d'accéder à l'interface d'administration exactement comme en local, sans exposer les services d'administration directement sur la liaison montante.

7.5 Authentification centralisée par serveur RADIUS

Rôle fonctionnel : centraliser l'authentification des usagers auprès d'un serveur dédié, plutôt que sur la base de comptes locale du routeur, ouvrant la voie à une gestion mutualisée entre plusieurs routeurs.

Cette piste a été testée au cours du projet, mais écartée de la configuration finale : l'activation de l'authentification RADIUS sans disponibilité garantie et rapide du serveur associé a introduit des délais d'authentification significatifs, dégradant l'expérience usager. Une mise en œuvre future nécessiterait de sécuriser la disponibilité et la latence du serveur RADIUS avant réactivation.

7.6 Autres pistes d'amélioration réseau

- Liaison de secours (LTE) prenant automatiquement le relais en cas de coupure de la liaison montante principale
- Système de vouchers à usage unique, adapté à un modèle de vente par ticket prépayé
- Sélecteur de langue locale supplémentaire sur les pages du portail captif
- Tableau de bord de supervision unifiant l'état du réseau et l'état de l'alimentation électrique du site

8. Conclusion

La configuration décrite dans ce document constitue une solution de hotspot Wi-Fi complète, opérationnelle et reproductible, répondant aux exigences d'accès utilisateur, d'administration du réseau et de maintenance automatisée que nous nous étions fixées. Les évolutions présentées en dernière partie tracent les pistes d'amélioration que nous jugeons pertinentes pour la poursuite de ce travail.